

Odluke u programu

Dijagram tijeka:



Oblik naredbe odluke:

Mali engleski rječnik:

if – ako

true – istina

false – laž

TEST logički uvjet

IFTRUE [naredbe koje se izvršavaju ako je uvjet istinit]

IFFALSE [naredbe koje se izvršavaju ako je uvjet lažan]

Uvjet koji se u dijagramu toka pokazuje romбом u Logu se piše nakon **TEST** naredbe.

Naredbe koje se izvršavaju ako je uvjet **istinit** slijede u naredbi **IFTRUE (IFT)**

– pišu se u uglatoj zagradi []

Naredbe koje se izvršavaju ako je uvjet **lažan** slijede u naredbi **IFFALSE (IFF)**

– pišu se u uglatoj zagradi []

Uvjet uvijek mora biti takav da se na njega može odgovoriti sa **istina** ili **laž**. Takve uvjete zovemo **logički uvjeti**.

Primjeri uvjeta:

:a = 10	jednako
:a > 10	veće
:a < 10	maje
:a <= 10	manje ili jednako
:a >= 10	veće ili jednako
:a <> 10	različito

1. Zadatak:

Napravite program JEDNAKI :A :B koji unosi 2 broja i ispituje jesu li brojevi jednaki

Primjer:

```
TO jednak :A :B
TEST :A = :B
IFT [ PR [brojevi su jednaki]]
IFF [ PR [ brojevi nisu jednaki]]
END
```

Poziv programa:

```
jednak 12 12
brojevi su jednaki

jednak 12 67
brojevi nisu jednaki
```

2. Zadatak:

Napravite program ISPIT :A koji unosi broj i ispituje je li broj manji, veći ili jednak 100.

prvi uvjet: je li broj manji od 100: ako je uvjet istinit ispisujemo poruku da je manji,
ako uvjet nije istinit postavljamo

drugi uvjet: ispituje li broj veći od 100: ako je uvjet istinit ispisujemo poruku da je veći,
ako uvjet nije istinit ispisujemo poruku da je jednak 100.

```
to ispit :a
TEST :a<100
IFT [ PR [ manji je od 100 ] ]
IFF [ TEST :a>100
      IFT [ pr [ veći je od 100 ] ]
      IFF [ pr [ broj je jednak 100 ] ] ]
End
```

3. Zadatak

Napravite program VODORAVNO :A :B koji crta pravokutnik tako da mu je dulja stranica uvijek vodoravno, a kraća okomito.

```
to pravo :a :b
repeat 2 [ fd :a rt 90 fd :b rt 90 ]
end
```

:A

:B



```
to vodoravno :a :b
test :a > :b
ift [ pravo :b :a ]
iff [ pravo :a :b ]
end
```

4. Zadatak: Program CRTA :N :X za zadani lik sa :n stranica duljine stranice :x. ako unesemo N manji od 3 daje poruku da lik mora imati više od 2 stranice.

```
to lik :n :x
repeat :n [ fd :x rt 360/:n ]
end
```

```
to crta :n :x
test :n<3
ift [ pr [ Lik mora imati više od 2 stranice ] ]
iff [ lik :n :x ]
end
```

Drugi oblik ispitivanja uvjeta

IF logički uvjet [naredbe]

IF logički uvjet [naredbe se izvode **ako je uvjet istinit**]

PR [Program ide dalje...]

Naredbe u zagradi se izvode samo ako je uvjet istinit, ako uvjet nije istinit, program naredbe u uglatoj zagradi neće izvršiti i program samo nastavlja dalje

Treći oblik ispitivanja uvjeta

Istina laž
 ↓ ↓

IFELSE logički uvjet [naredbe1] [naredbe2]

IFELSE logički uvjet [naredbe za **istinit** uvjet] [naredbe za **lažan** uvjet]

Ako je uvjet istinit, izvršit će se naredbe iz prve zagrade, ako uvjet nije istinit izvršit će se naredbe iz druge zagrade

5. Zadatak: Program GODINE :G unosi godine, ispituje je li osoba punoljetna i ispisuje poruke.

```
to godine :g
ifelse :g < 18 [ pr [ Ti nisi punoljetan ] ] [ pr [ Ti si punoljetan ] ]
end
```

6. Zadatak: Program USPJEH :X unosi brojčanu ocjenu, a ocjenu koja je upisana ispisuje riječima.

```
to uspjeh :x
if :x=5 [ pr "odličan ]
if :x=4 [ pr "vrlo_dobar ]
if :x=3 [ pr "dobar ]
if :x=2 [ pr "dovoljan ]
if :x=1 [ pr "nedovoljan]
end
```

Logički operatori OR i AND (povezivanje uvjeta)

```
to ocjena :x
if (or :x<1 :x>5 ) [ pr "nije_ocjena]
if :x=5 [ pr "odličan ]
if :x=4 [ pr "vrlo_dobar ]
if :x=3 [ pr "dobar ]
if :x=2 [ pr "dovoljan ]
if :x=1 [ pr "nedovoljan]
end
```

UNOS podataka tijekom izvođenja programa

Naredba RW – read word (učitavanje riječi)

MAKE "X RW program staje i čeka unos, ono što unesemo sprema se naredbom MAKE u zadanu varijablu (X)

```
to rukomet
PR [ Je li Hrvatska pobijedila Dansku? ]
PR [ UPUTA: Za pobjedu unesi 1, za poraz 0 ]
MAKE "D RW
PR [ ]
PR [ Je li Katar pobijedio Argentinu ? ]
PR [ UPUTA: Za pobjedu unesi 1, za poraz 0 ]
MAKE "A RW
PR [ ]
TEST (AND :D=1 :A=1)
IFT [ PR [ Idemo dalje !!!] ]
IFF [ PR [ Ne prolazimo dalje ] ]
end
```

Primjer programa koji ispisuje pitanje i nudi nekoliko odgovora. Program se zaustavlja i čeka da korisnik izabere odgovor. Odgovor se upisuje u varijablu x. Program ispituje je li odgovor točan, ako je točan ispisuje poruku i dodaje jedan bod u brojač. Brojaču (z) na početku programa moramo dati vrijednost 0. Ako odgovor nije točan, program ispisuje poruku.

```
to kviz
MAKE "z 0
PR [ Pitanje 1.]
PR [ Koja naredba u Logu pokreće kornjaču 50 koraka natrag? ]
PR [ a) FD 50]
PR [ b) BK 50]
PR [ c) CS 50]
PR [ d) RT 50]
PR [ ]
PR [ Unesi a, b, c ili d]
MAKE "x RW
PR SE [ Tvoj odgovor je] :x
PR [ ]
TEST :x = "b
IFT [ PR [ Odgovor je točan] MAKE "z :z + 1 ]
IFF [ PR [ Odgovor nije točan] ]
PR ( SE [ Do sada imaš] :z "bodova )
end
```